Химический факультет МГУ

Лаборатория химической кибернетики

Исаева Е.В.

MAPLE 

Пакет вычислительной и символьной математики

Обзорная экскурсия

(Методическое руководство к спецпрактикуму 

“Математические методы в химии”)
2010г.

Содержание. (кликабельно)

Начнем, пожалуй...
Worksheet mode and Document mode
Настройка для работы по умолчанию в Worksheet mode 1-D Math input.
О режиме ввода 2-D Math Input и палитрах
Ввод операторов, присвоение значений переменной
Пакеты
Вычисления
Системные переменные, константы и некоторые обозначения
Преобразование выражений
Решение уравнений
Дифференцирование, интегрирование
Решение дифференциальных уравнений
Поиск экстремума функции
Построение графиков
Аппроксимация полиномами и сплайнами
Разложение функций в ряд Тейлора
Форматированный вывод на экран
Программирование в Maple
Вектора и матрицы
Работа с файлами данных
Начнем, пожалуй...

Программный  продукт Maple принадлежит к классу систем компьютерной  алгебры. Это мощная математическая система, способная проводить как численные, так и символьные преобразования (вычисления) и имеющая хорошо проработанный пакет научной графики. В программе MAPLE прекрасный файл помощи. Вам придется часто им пользоваться для уточнения синтаксиса операторов. Вызывается через меню или по кнопке F1, когда курсор стоит на интересующем вас операторе. Очень полезно  внимательно разбирать приведенные примеры.

   Вычисления в программе ведутся в режиме диалога: ввел исполняемый оператор – нажал Enter – получил ответ. Несколько исполняемых (текстовых) строк могут быть объединены в секции, которые можно свернуть  ([image: image1.png]


) или развернуть ([image: image2.png]


),  кликнув на  значок.
ЭТО ВАЖНО!
        Помните, что результаты вычислений зависят от последовательности выполненияоператоров, а не от последовательности, в которой Вы написали их на листе. Если нужно изменить последовательность операторов, скопируйте и перенесите в нужное место, иначе Вам будет крайне трудно проанализировать результаты вычислений или найти ошибку.       

 Оператор restart; начинает новый сеанс работы, при этом рабочее поле не изменяется, но все результаты выполненных до рестарта операторов "забываются".
К оглавлению
Worksheet mode and Document mode - 2 типа файлов

При работе в Maple могут быть созданы файлы двух видов: документ (Document mode) и рабочий лист (Worksheet mode). Работа в Document mode дает большие возможности редактирования, позволяет создать полностью готовый к печати документ. Математические же функции одинаково доступны в обоих случаях. При создании нового файла при выборе пунктов меню File – New предоставляется возможность выбора вида документа: Worksheet mode и Document mode. Поскольку нашей задачей не является подготовка к печати документов, мы постоянно будем работать в Worksheet mode. 
Два режима ввода: 2-D и 1-D. 
 В  Worksheet mode возможны два режима ввода информации: 2-D и 1-D Math input. На рисунке приведены результаты ввода в том и другом режиме операторов дифференцирования и интегрирования. Переключение между режимами осуществляется кнопками Text - Math панели инструментов. Каждый режим имеет свои плюсы и минусы. 

Мы будем работать в режиме1-D Math input  (Подсвечена кнопка Text на панели инструментов). 
К оглавлению
Настройка для работы по умолчанию в Worksheet mode 1-D Math input.

1) Tools - Options вкладка General

       Уберите галочку в строке Auto save (нижняя строка)

2)    ...    вкладка Display    

          Строка Input display - выберите Maple Notation

3)    ...    вкладка Interface   

          Строка Default format for new worksheets  -    выберите  Worksheet 

          Уберите галочку в строке Startup dialog

4)    Примените эти установки глобально, т.е. нажмите кнопку Apply Globally  внизу окна.

К оглавлению
О режиме ввода 2-D Math Input и палитрах
Режим ввода 1-D является наиболее быстрым: так для ввода уравнения  

 в этом режиме набираем: x^2/a+b*y^3/z^2=z. В режиме 2-D, т.е. "двумерном", результат ввода будет иметь стандартный математический вид, но за это придется заплатить дополнительными нажатиями клавиатуры: каждый переход из показателя степени или знаменателя должен быть отмечен нажатием клавиши "стрелка вправо". Т.е. мы должны будем набрать x^2[image: image5.wmf]/a[image: image6.wmf]+b*y^3[image: image7.wmf]/z^2[image: image8.wmf]. Собственно, дело не столько в дополнительных нажатиях, сколько в том, что при быстром вводе нажатие на стрелку довольно часто пропускают, что приводит к ошибкам: часть выражения вводится в показатель степени или в знаменатель. Как в режиме 1-D, так и в режиме 2-D можно использовать палитры. На рисунке красными овалами показаны треугольники, с помощью которых можно вызвать на экран или убрать список палитр. В приведенном примере открыта палитра выражений. Выбор значка определенного интеграла в режиме 1-D приведет к печати в  исполняемой строке соответствующей команды int(f,x=a..b); в режиме 2-D - к печати выражения в привычном математическом виде [image: image9.wmf]Нужные значения  a, b, f, x подставляют в полученное выражение так же, как в Microsoft Equations. Впрочем, использование палитр резко замедляет работу, так что гораздо разумнее потратить некоторое время на освоение синтаксиса основных команд.

Есть в режиме 2-D интересная возможность. Если в исполняемую строку ввести выражение, уравнение, то нажатием правой кнопки мыши вызывается контекстное меню, из которого можно выбрать операцию, которую желательно выполнить. Например, дифференцировать, интегрировать выражение, решить уравнение и т.д. На рисунке  - выбор команды "продифференцировать по x" для введенного выражения x3y.

К оглавлению
Ввод операторов, присвоение значений переменной

Для вычислений вводим выражения (операторы) в исполняемую строку  (начинается c  >). Вводимые нами операторы печатаются красным цветом. Конец оператора обозначает точка с запятой или двоеточие (в том случае, если нужно, чтобы действие было выполнено, но на экран результат не выдавался). В одной строке можно вводить несколько операторов. Оператор выполняется после нажатия клавиши Enter, когда курсор стоит на строке оператора (необязательно в конце выражения). Результат выполнения операторов печатается синим цветом. Комментарий можно добавить в строке исполняемого оператора после знака #.  Для записи своих выводов или анализа вычислений перейдите в текстовый  режим. (Кнопка T  панели инструментов).

>x:=2; y:=x^2: z:=x^2+y^3;# пример операторов присвоения.
x:=2

z:=68
 «Очистить» переменную, т.е. отменить сделанное  ранее присвоение значения можно следующим образом: 
>y:='y';                   y:=y      
Синтаксис ввода алгебраических выражений в MAPLE стандартен для языков программирования:

+, -, *, /, ^ - знаки сложения, вычитания, умножения, деления, возведения в степень соответственно. Знак * между сомножителями обязателен.  

             Некоторые функции:

                  квадратный корень -  sqrt(x)

                  тангенс - tan(x)

                  арккосинус -  arcos(x) 

                  логарифм натуральный -  ln(x)

                  логарифм по основанию b (в том числе,  десятичный) -  log[b](x)

                  бесконечность - infinity
                  модуль (абсолютная величина) числа abs(x)
К оглавлению
Пакеты
Расширить список доступных в ядре программы Maple операторов можно при подключении специализированных пакетов, таких как stats, plottools, LinearAlgebra, Physics и т.д.
Подключить нужный пакет можно с помощью оператора with(название пакета); 
при этом выдается список доступных в этом пакете операторов:

> with(stats);

                      [anova, describe, fit, importdata, random, statevalf, statplots,transform]

Если хотите подавить выдачу на экран списка операторов пакета (в некоторых пакетах этот список весьма длинен), закончите оператор вызова пакета двоеточием:  
> with(LinearAlgebra):
К оглавлению
Вычисления.

Перевести простую дробь в десятичную, подставить численное значение константы можно с помощью оператора evalf.  В этом же операторе можно задать число значащих цифр, с которым ведется вычисление в данном операторе.
> x:=3/4; 

> y:=sin(Pi/6);

> evalf([x,y]);
[0.7500000000, 0.5000000000]

> evalf([x,y],2);
[0.75, 0.50]

К оглавлению
Системные переменные, константы и некоторые обозначения
Pi  -   число π

ln(x) –  натуральный логарифм

log[b](x) - логарифм по основанию b
I    -  мнимая единица 

infinity  -   I (бесконечность)
sqrt(x)  -  квадратный корень

tan(x) -   тангенс

arcos(x) -  арккосинус 
Системная переменная Digits задает число значащих цифр (не знаков после запятой!), с которыми будут проводиться вычисления в сеансе работы после выполнения присвоения Digits нового значения и до его отмены. 

По умолчанию Digits=10.

> X:=0.053708642934;
X:=0.053708642934
> Digits:=3;    
Digits:=3

> evalf(X);
0.0537

% -  Если необходимо произвести некоторое действие с только что полученным результатом, не обязательно его переписывать (копировать) в выражение для оператора. Вместо него можно поставить значок %:
> solve(cos(2*x)=1/2,x);
[image: image14.wmf]p

> evalf(%);
0.5235987758

EnvAllSolutions
Эта системная переменная обеспечивает поиск всех периодических решений при значении true и  лишь главных решений при значении false (по умолчанию)
> restart: solve(cos(x)=1/2,x); 
> _EnvAllSolutions:=true; 
> solve(cos(x)=1/2,x);

3

1


_EnvAllSolutions :=true

Z1 обозначает целое число, B1- бинарное число (0 или 1). 

К оглавлению
Преобразование выражений
[image: image71.wmf])

Для упрощения выражений  можно воспользоваться оператором “simplify”:

Упростить выражение  

>simplify((1/p^4-1/q^4)/(p+q)^3+2*(1/p^3-1/q^3)/(p+q)^4+2*(1/p^2-1/q^2)/(p+q)^5);


[image: image19.png]



Упростить выражение  

>simplify((z^3*(x+1))^(1/3));

(z3(x+1))1/3
> simplify((z^3*(x+1))^(1/3),symbolic);

z(x+1)1/3
Разложить на множители можно с помощью оператора “factor”:

>factor((a+b+c)^3-(a+b-c)^3-(b+c-a)^3-(c+a-b)^3);

24abc
Подставить в выражение вместо переменной его значение можно оператором “subs”, а раскрыть скобки -  оператором “expand”:

Доказать, что:        a(s-a)2+b(s-b)2+c(s-c)2+2(s-a)(s-b)(s-c)=abc,     если     2s=a+b+c.

>expand(subs(s=(a+b+c)/2,a*(s-a)^2+b*(s-b)^2+c*(s-c)^2+2*(s-a)*(s-b)*(s-c)-a*b*c));

0

Полезным является оператор объединения “combine” с различными опциями:
>combine(exp(ln(x)+3*ln(y)),exp);

x y3

combine(x^(1-n)*x^(m+3),power);

x4-n+m
>combine(16*cos(x)^3-12*cos(x)-16*sin(x)*cos(x)^3+8*sin(x)*cos(x),trig);

4cos(3x) - 2sin(4x)

К оглавлению


Решение  уравнений. 

Решая уравнение относительно переменной, Maple выводит результат на экран, но не присваивает значение переменной, если такое присвоение не указано в операторе: 

> solve(x-5=1,x);      x;
6

x
> y:=solve(y+2=-1,y);     y;
y:=-3

-3

При решении нелинейных уравнений и систем уравнений формируется список корней. Из этого списка можно извлечь отдельный корень для дальнейших вычислений:
> s:=solve(x^3-10*x^2+31*x-30,x);
s:=2,3,5

> x:=s[3];

x=5

При решении систем уравнений каждый член списка решений представляет собой равенство и  для выделения отдельного корня необходимо использовать оператор выделения правой части - rhs. (Существует также оператор lhs - выделение левой части равенства). В ряде случаев списки корней могут быть двумерными.  
> s:=solve({x*y=3, 2*(x+y)=8},{x,y});
s:={x=3,y=1}, {y=3,x=1}

> length:=rhs(s[1,1]);   width:=rhs(s[1,2]);
length:=3

width:=1

Элементы двумерного списка:
> s[1];  s[1,1]; rhs(s[1,1]);
{x=3,y=1}

x=3

3

> s[1,2];
    y=1 
Если результат решения представлен через функцию RootOf, получить все корни можно, используя функцию allvalues. 
> solve( {x^2*y^2-b, x^2-y^2-a}, {x,y} );

{y=RootOf(_Z4+_Z2a-b),  x= RootOf(_Z2-RootOf(_Z4+_Z2a-b)2-a)}

> allvalues({%});   # Вот все 8 решений системы уравнений:

К оглавлению
Дифференцирование, интегрирование
Для получения производной используют команду diff(expr, x,$n), где n – порядок производной. Команда вычисления смешанной производной включает все переменные, по которым дифференцируем, с указанием порядка.


[image: image22.png]> diff (x*3*1n(y) ,x$2,y) ;7




Для вычисления интегралов используется функция  int(expr,x). Для вычисления кратных интегралов  необходимо использовать вложенные функции. Для вычисления определенных интегралов укажите пределы интегрирования. 

[image: image23.png]> int(sin(x) ,x) ;
—cos(x)

> int (int (int(x"2*y/z,2) ,y) ,%) ;
%xsyz In(z)

> int (x"2*exp (-x"2) ,x=-infinity..infinity) ;

3=




К оглавлению
Решение дифференциальных уравнений
Оператор dsolve решает дифференциальные уравнения и системы дифференциальных уравнений. Начальные (граничные) условия задаются в том же операторе:
Решить уравнение     

>dsolve(diff(y(x),x)-y(x)=x*sin(x),y(x));

y(x)= -1/2 xcos(x)-1/2 cos(x ) –1/2 xsin(x) +ex_C1
Решить систему уравнений   

   при начальных условиях  y(0)=0, z(0)=1.

> dsolve({diff(y(x),x) = 2*z(x) - y(x) - x,  diff(z(x),x) = y(x) + x,y(0)=0,z(0)=1},{y(x),z(x)});

{ y(x) = ex - e-2x – x,  z(x) = ex + 1/2 e-2x – 1/2 }


Решить уравнение
   

при граничных условиях:  y(0) = 1;  y(1) = 0

> eq:=diff(y(x),x$2)-y(x)=x*sin(x);    # Определяем уравнение


[image: image26.png]



> ss:=dsolve({eq,y(0)=1,y(1)=1},y(x));

> evalf(ss,4);

y(x) = 1.016e-x + 0.4847ex – 0.5000cos(x) – 0.5000xsin(x)

К оглавлению
Поиск экстремума функции
> with(Optimisation):     # Подключаем пакет оптимизации
> F:=x^2-x-2;   F1:=-x^2+x+2;

F:=x2 – x – 2

F1:= -x2 + x + 2
> Minimize(F);  Maximize(F1);

[-2.2500000000000, [x = 0.499999999999999944]]

[2.2500000000000, [x = 0.499999999999999944]]

Для более сложных функций лучше указать начальное приближение для аргумента:
> Minimize(F,initialpoint={x=-2}); 

[-2.2500000000000, [x = 0.499999999999999956]]

К оглавлению
Построение графиков
При построении графика можно задать интервал, на котором строится график, цвет, стиль (линия, точки),  толщину линий. Для точечного графика можно задавать символы  (кружки, крестики, ромбики). Многие параметры можно изменять и после построения графика при помощи всплывающего меню. 

Для построения нескольких графиков в одной системе координат список функций записывают в квадратных скобках, разделяя запятыми.

>plot([x-2,2*x+1],x=-10..10,y=-12..12,color=[magenta,green],

style=[line,point],symbol=cross,title='Example',thickness=3);
[image: image28.png]Example

При построении нескольких графиков сложных функций можно сначала создать структуры графиков, а затем провести построение с помощью команды plots[display]. Тот же прием применяют, если на одном рисунке нужно построить графики нескольких функций, определенных на различных интервалах, в полярных и декартовых координатах. 
>g1:=plot([-0.5/cos(t),0.5/cos(t)],t=-Pi/4..Pi/4,coords=polar,

color=[magenta,magenta],thickness=3): 
> g2:=plot([-0.5,0.5],x=-0.5..0.5,color=[magenta,magenta],thickness=3):
> plots[display]({g1,g2});

[image: image29.png]



Если необходимо показать на графике дискретные точки, необходимо подключить пакет (plots) или пакет статистики (stats). После этого конструируем структуру и выводим на экран, например, совместно с аппроксимирующей функцией:

Построение графика интерполяционной функции и экспериментальных точек.
Экспериментальные точки X=[-10, -6, -2, 2, 6, 10];  Y=[0.28,0.47,0.72,1.30,2.12,3.49] интерполировать функцией exp(x/8). Построить график функции с экспериментальными точками.

1. Создать массивы координат экспериментальных точек:

> X := [-10, -6, -2, 2, 6, 10]; Y := [.28, .47, .72, 1.30, 2.12, 3.49];
[image: image30.wmf]
[image: image31.wmf]
2. Создать структуру графика функции:

> g1 := plot(exp((1/8)*x)):
3. Создать структуру графика экспериментальных точек (с помощью пакета stats).

> with(stats):

> g2:=statplots[scatterplot](X,Y,color=green,symbol=solidcircle,

  symbolsize=14):
Или с помощью пакета plots:

> with(plots):
 g3:=pointplot({seq(X[i],Y[i],i=1..6)},symbol=circle, symbolsize=24):
4. Показать на графике все структуры:

>plots[display](g1,g2,g3);

[image: image32.png]-1

Построение трехмерных графиков обеспечивает функция  plot3d.

> plot3d(sin(2*x+4*y), x=-1..1, y=-1..1);


[image: image33.png]



Графики могут быть построены в разных системах координат, например в сферической:

> plot3d(1.3^x*sin(y),x = -1..2*Pi, y = 0..Pi, coords=spherical, style=patch);


[image: image34.png]



Можно использовать переменные границы интервала построения по одной из осей: 
> plot3d(sin(x*y), x = -Pi..Pi, y = -x..x);


[image: image35.png]



К оглавлению
Аппроксимация полиномами и сплайнами
Для аппроксимации экспериментальных данных,  таблично заданных функций полиномами и сплайнами необходимо загрузить библиотеку «spline». Для аппроксимации полиномом используют оператор  interp(X,Y,x), для аппроксимации кубическими или квадратичными сплайнами – операторы  spline(X,Y,x,cubic) или  spline(X,Y,x,quadratic) соответственно. Ниже приведен пример аппроксимации «экспериментальных точек»  полиномом, кубическим и квадратичным сплайнами, построение структур для полинома, сплайнов, дискретных точек и вывод графика на экран:


> readlib(spline): X:='X': Y:='Y': X:=[0,1,2,3,4,5]: Y:=[0,1,4,3,2.5,1]:
> f1:=evalf(interp(X,Y,x),6); # Полином

f1:=-0.137500 x5 +1.81250 x4 -7.925000000 x – 8.43750 x3 + 15.6875 x2

>f2:=evalf(spline(X,Y,x,cubic),6); # Кубический сплайн 

                  0.162679x + 0.837320 x3  ,                                                       x < 1

                 -1.674641 + 2.67464x + 2.51196(x-1)2 – 2.18660(x-1)3 ,         x < 2

  f2:=         1.72249 + 1.13876x – 4.04785(x-2)2 + 1.90909(x-2)3  ,           x < 3

                  6.68899 – 1.22966x + 1.67943(x-3)2 – 0.94976(x-3)3 ,             x < 4

                 5.38038 – 0.720096x – 1.16986(x-4)2 + 0.389952(x-4)3,          otherwise

> f3:= evalf(spline(X,Y,x,quadratic),6);  # Квадратичный сплайн
                  0.756939 x2  ,                                                                         x < 0.500000

                 -1.48612 + 2.48612x + 1.72918(x-1)2,                                   x < 1.50000

  f3:=         1.83347 + 1.08326x – 3.13204(x-2)2 ,                                   x < 2.50000

                  5.95711 – 0.98570x + 1.06308(x-3)2 ,                                   x < 3.50000

            7.17620 – 1.16905x –1.24643(x-4)2 ,                                    x < 4.50000

             1. + 2.41547 (x-5)2  ,                                                            otherwise
Построение структур полинома и сплайнов:
> q1:=plot([f1,f2,f3],x=0..6,y=-2..6,color=[green,maroon,blue]):

> with(stats)

>q2:=statplots[scatterplot](X,Y,color=red,symbol=circle,symbolsize=20):

Построение графика на экране:
> plots[display]([q1,q2]);  
[image: image36.png]




К оглавлению
Разложение функций в ряд Тейлора

Разложим в ряд Тейлора sin(x) в окрестности точки x=0, ограничив разложение степенью x5: 
> taylor(sin(x),x=0,5);
x – 1/6 x3 + O(x5)
 Для  дальнейшей работы с полученным рядом (например, для построения графика, вычисления значения) нужно преобразовать ряд Тейлора в полином:
> t4:=convert(%,polynom);
t4:=x – 1/6 x3
> evalf(subs(x=Pi/6,t4));
0.4996741796

> plot([sin(x),t4],x=0..Pi);
[image: image37.png]



К оглавлению
Форматированный вывод на экран
Оператор printf("format", list of output);
Пример вывода приведен ниже. Здесь использованы следующие форматы:

g - выводит десятичное число в е-форме либо c десятичной точкой в зависимости от его значения

f - выводит десятичное число в форме c десятичной точкой (по умолчанию - шесть десятичных знаков)

d - целое

a - алгебраическое выражение
\n - перевод строки
Подробнее о форматах вывода см. Help.
> delta:=7.73705e-09: a:=0.007: beta:=0.0245009:

n:=3: t[2]:=5*x^2-11*y+lambda:

printf(" delta=%g a=%g beta=%f\n tayl[%d]=%a",delta,a,beta,n-1,t[n-1]);
 delta=7.73705e-09 a=0.007 beta=0.024501

 tayl[2]=5*x^2-11*y+lambda

К оглавлению
  Программирование в Maple

Операторы цикла

В пакете Maple есть несколько операторов цикла: 
1) for i from  Nbegin by Step to Nend do … end do: 
> for i from 1 to 3 do N[i] := i*(i+1)^2 end do;
N1:=4

N2:=18

N3:=48

2) for i from  Nbegin by Step while i<Nend   do … 
> tot := 0:
> for i from 6 by 3 while i <= 100 do
>    tot := tot + i:
> end do:
> print(tot);
1680
3) оператор, в котором переменная цикла пробегает все значения из заданного списка: 
           for   i    in    a,b,c,d,e,f     do    …     end do; 

> tot := 1:
> for z in 1, x, y, q^2, 3 do
>   tot := tot*z end do:
> print(tot);
3xyq2
Условный оператор  имеет  вид: if (условие) then … <elif(условие)  then ... > else … end if;  

(в угловых скобках необязательная часть)
> n:=3:

if n < 100 then n else n^2 end if;
3
> n:=20:

if n < 100 then "Мало!" elif n<150 then "Как раз!" else "Много!!!" end if;
“Мало!”
> n:=120:

if n < 100 then "Мало!" elif n<150 then "Как раз!" else "Много!!!" end if;
“Как раз!”
> n:=200:

if n < 100 then "Мало!" elif n<150 then "Как раз!" else "Много!!!" end if;
“Много!!!”
Процедуры

Написать процедуру расчета длины и ширины прямоугольника по площади и периметру.
> lw:=proc(S,P)               # заголовок процедуры
local l,h,eqns,A;              # объявление переменных
eqns:={l*h=S, 2*(l+h)=P};      # задание уравнений
A:=solve(eqns,{l,h}):          # решение системы уравнений
if rhs(A[1,1])<rhs(A[1,2]) then  

print(length=rhs(A[1,2]),    width=rhs(A[1,1]))

else print(length=rhs(A[1,1]),    width=rhs(A[1,2]))

end if:  # присваиваем бОльшее значение переменной l(длина), меньшее - h(ширина)
> end proc:
Расчет ( вызываем процедуру с фактическими параметрами S=21, P=20):
> lw(21,20);

length=7,  width=3

Процедура формирования линейной комбинации аргументов:

> lc := proc( s, u, n, t, v )

        description "form a linear combination of the arguments";

        s * u + n*t * v

end proc:
> lc( Pi, x, 5,-I, y );
x – 5Iy
К оглавлению
Вектора и матрицы
Для работы с векторами и матрицами подключим пакет LinearAlgebra
> with(LinearAlgebra):

и затем используем команды пакета.

Ввод матриц
> A:=Matrix([[6,-3,4],[8,9,14]]);

Или так:
> <<6|-3>,<8|9>>;

Если Вам удобнее вводить данные по столбцам, это можно сделать так:
> <<6,8>|<-3,9>>;

Выделить часть матрицы можно командой SubMatrix:
> B:=SubMatrix(A,1..2,2..3);

Ввод векторов
Вектор-столбец вводят командой 
> Vector([7,-3,2]);

Или так:
> <7,-3,2>;

А вектор-строку
> Vector[row]([11,27,6]);
[ 11  27  6 ]
Или так:
> <11|27|6>;
[ 11  27  6 ]
Суммы, произведения векторов

> u := Vector([1,2,3]); v := Vector([6,5,7]);

[image: image44.wmf]ú

ú

ú

û

ù

ê

ê

ê

ë

é

=

3

2

1

:

u



[image: image45.wmf]ú

ú

ú

û

ù

ê

ê

ê

ë

é

=

7

5

6

:

v


> u+v;

Скалярное произведение:
> DotProduct(u,v);
37
> u.v;
37
Векторное произведение:
> CrossProduct(u,v);

> u &x v;

Суммы, произведения матриц
> A := Matrix([[1,3],[5,7]]); B := Matrix([[1,1],[1,1]]);

> A+2*B;

> A.B;

> A.u;

Обратная матрица
> A:=<<a,c>|<b,d>>;

> MatrixInverse(A);

> A^(-1);

Транспонирование:
> Transpose(A);

> A^%T;

Собственные вектора и собственные значения:

> M:=Matrix([[4, -1, 6],[2, 1, 6], [2, -1, 8]]);

> Eigenvectors(M);

Можно сделать выдачу более подробной:
> Eigenvectors(M,output='list');

В этом списке первое число - собственное значение, его кратность, затем в фигурных скобках - собственный вектор (вектора).
Длина вектора, угол между векторами:

> V1:=<1,5,3>;    LV1:=VectorNorm(V1,2);

> V2:=<4,2,-3>;    LV2:=VectorNorm(V2,2);

[image: image64.wmf]ú

ú

ú

û

ù

ê

ê

ê

ë

é

-

=

3

2

4

:

2

V



[image: image65.wmf]29

> angleV1V2:=evalf(180*VectorAngle(V1.V2)/Pi);
angleV1V2:=80.9706
К оглавлению
Работа с файлами данных

Открытие/закрытие файла
Перед началом работы с файлом его рекомендуют открыть для чтения, записи или дополнения данных оператором fopen(fileName, acсessMode, fileType). Здесь  fileName - имя файла (с указанием пути), acсessMode - тип доступа: для чтения (READ), записи (WRITE) или дополнения данных (добавления данных в конец существующего файла) (APPEND). На практике эта процедура необходима обычно лишь для  acсessMode = APPEND. В других случаях файл открывается (если существовал) или создается (для acсessMode = WRITE) в момент первого обращения к файлу. Тип файла может иметь значение TEXT  или BINARY. Мы рассмотрим некоторые возможности работы с файлами TEXT, содержащими текстовую и числовую информацию. Для этих файлов опция fileType может быть опущена.
Пример:
> fopen("S:\\o13\\Teach\\file1.txt",READ);
Для сокращения записи обычно вводят дискриптор файла. 
 Если используем оператор открытия файла:
> fd:=fopen("S:\\o13\\Teach\\file1.txt",READ);
Eсли читаем или пишем в файл, то предварительно объявляем дискриптор:
> fd:="S:\\o13\\Teach\\file1.txt";
тогда вместо длинной строки в двойных кавычках в операторах чтения-записи пишем просто fd.

Чтение из файла
Рассмотрим 2 варианта чтения из файла: самый простой и наиболее универсальный.
ImportMatrix
Самым простым вариантом считывания текстовой и числовой информации, записанной в виде прямоугольной матрицы (т.е. когда формат записи всех строк одинаков), является использование Ассистента для импорта данных.  

Пример: считаем информацию из файла S:\\o13\\Teach\\file1.txt. В файле записаны "координаты атомов".  В меню выбрать Tools - Assistants - Import Data... В появившемся окне выбрать файл для чтения и формат файла (для форматов *.txt, *.ent, ...   delimited).

После нажатия кнопки "next" появится следующее окно. На большом белом поле видно содержание выбранного файла. Выбираем тип данных Data Type = anything, Разделитель Separator = Space Separator (пробел) и нажимаем "Done". На рабочем листе появляется команда и результат ее выполнения:

> ImportMatrix("D:\\file1.txt", source=delimited, delimiter=" ", format=rectangular, datatype=anything, transpose=false, skiplines=0);

Если необходимо присвоить имя считываемому массиву данных, придется подставить слева оператор присвоения и снова выполнить чтение: 

> A:=ImportMatrix("D:\\file1.txt", source=delimited, delimiter=" ", format=rectangular, datatype=anything, transpose=false, skiplines=0);

Теперь данные доступны для работы: 
> A[1,2]:=A[1,2]-1; print(A);
A1,2:=9.23

Для чтения данных из таблицы Excel нужно указать имя файла и интервал ячеек считываемых данных:
> ExcelTools:-Import("D:\\fileex.xls", "Лист1", "B10:E14");

[[“x”,”y”,”yрасч”,”y-yp”], [0.00366300366003663,-36.106088431522274,-36.088539762159996,

-0.0175486693622],[0.003333333333333333335,-16.962395968369776,-16.991996477849767,

0.029600509479990933],[0.002857142857142857,10.580110363410927,10.591899377265037,

0.011789013854109243],[0.002680965147453083,20.7969385027598,20.797201329023324,

-0.00026282626352269745]]

Т.к. Maple выдает избыточное число десятичных знаков, возможно, потребуется округлить данные:

> C:=evalf(ExcelTools:-Import("D:\\fileex.xls", "Лист1", "B10:E14"),6);

C:=[[“x”,”y”,”yрасч”,”y-yp”], [0.00366300,-36.1061,-36.0885,0.0175487],

[0.00333333,-16.9624,-16.9920,0.0296005],[0.00285714,10.5801,10.5919,0.0117890],

[0.00268097,20.7969,20.7972,-0.000262826]]

Чтение с указанием формата.
В ряде случаев придется считывать файл, указывая формат записи, например,   оператором fscanf. Этот оператор возвращает список, где элементы разделены запятыми. Например, файл, содержащий таблицу данных: 
   1  2  3.1
   2  3  4.2
   3  4  5.3 
будет прочитан следующим образом:
> restart:

  fd1:="D:\\1.txt";

  a:=fscanf(fd1,"%d%d%f\n%d%d%f\n%d%d%f\n");
fd1:=”D:\1.txt

a:=[1,2,3.1,2,3,4.2,3,4,5.3]
а чтение первого элемента
> a1:=fscanf(fd1,"%d");
a1:=[1]
дает список с 1 элементом. Для того, чтобы a1, было присвоено значение 1 следует прочитать так:
> a1:=fscanf(fd1,"%d")[1];
a1:=1
Для получения массива данных A придется написать цикл:
> restart:

  A:=array(1..3,1..4):

  fd:="D:\\file1.txt":

  for i from 1 to 3 do

    A[i,1]:=fscanf(fd,"%6a")[1]:

    for j from 2 to 4 do

      A[i,j]:=fscanf(fd,"%7f")[1];

    end do:

  end do:

  print(A);

Запись в файл
    Для записи в файл воспользуемся форматным выводом. Запишем массив A в файл file1W.txt:
> fw:="D:\\file1W.txt":

  for i from 1 to 3 do

   fprintf(fw,"%3d  %6s", 1,A[i,1]):

    for j from 2 to 4 do

     fprintf(fw,"%7.2f",A[i,j]);

    end do;

   fprintf(fw,"\n");

  end do:

  fclose(fw);
А теперь дополним файл file1W данными о второй молекуле, смещенной относительно первой 

на вектор [10,10,10]:
> fopen(fw,APPEND):

  for i from 1 to 3 do

   fprintf(fw,"%3d  %6s", 2,A[i,1]):

    for j from 2 to 4 do

     fprintf(fw,"%7.2f",A[i,j]+10.);

    end do;

   fprintf(fw,"\n");

  end do:

  fclose(fw):
Получим такой файл:
1   Атом1  10.23   9.45  12.32
 1   Атом2  15.34  10.46  11.35
1   Атом3  16.23   8.45  15.32
 2   Атом1  20.23  19.45  22.32
 2   Атом2  25.34  20.46  21.35
2 Атом3  26.23  18.45  25.32
К оглавлению

